



Experiment #1: Flash An LED


The purpose of this experiment is to flash an LED with the BASIC Stamp. Flashing LEDs are often used as alarm indicators.

New PBASIC Elements/Commands:

- CON
- HIGH
- LOW
- PAUSE
- GOTO

Building The Circuit

All StampWorks experiments use a dashed line to show parts that are already on the NX-1000 board. The LED is available on the "LED MONITOR 16 CHANNELS" part of the board.



Since the StampWorks lab board has the LEDs built in, all you have to do is connect one to the BASIC Stamp.

1. Start with a six-inch (15 cm) white wire. Strip ¼-inch (6 mm) of insulation from each end.
2. Plug one end into BASIC Stamp Port 0.
3. Plug the other end into LED Monitor Channel 0

Experiment #1: Flash an LED

```
'
'
' File..... Ex01 - Blink.BS2
' Purpose... LED Blinker
' Author.... Parallax
' E-mail.... stamptech@parallaxinc.com
' Started...
' Updated... 01 MAY 2002
'
'   {$STAMP BS2}
'
' =====
'
' -----
' Program Description
' -----
'
' Blinks an LED connected to P0
'
' -----
' I/O Definitions
' -----
LEDpin          CON      0          ' LED connected to Pin 0
'
' -----
' Constants
' -----
DelayTime       CON      500        ' delay time in milliseconds
'
' -----
' Program Code
' -----
Main:
HIGH LEDpin          ' turn LED on
PAUSE DelayTime      ' pause for a bit
LOW LEDpin           ' turn LED off
PAUSE DelayTime      ' pause while off
GOTO Main            ' do it again

END
```

Behind The Scenes

Each of the Stamp's I/O pins has three bits associated with its control. A bit in the `DIRS` word determines whether the pin is an input (bit = 0) or an output (bit = 1). If the pin is configured as an output, the current state of the pin is stored in the associated bit in the `OUTS` word. If the pin is configured as an input, the current pin value is taken from the associated bit in the `INS` word.

`HIGH` and `LOW` actually perform two functions with one command: the selected pin is configured as an output and the value is set in the `OUTS` word (1 for `HIGH`, 0 for `LOW`).

For example, this line of code:

```
HIGH 0
```

performs the same function as:

```
Dir0 = 1           ' make Pin 0 an output
Out0 = 1           ' set Pin 0 high
```